Overview of the Type System of PVS

Lecture Interactive Proof Tools Summer Term 2003

Hans de Nivelle, Patrick Maier

Types in Programming Languages

• provide structure

– provide specification which can be checked (type checking), – provide specification which documents, • are naturally associated with functions, • ensure certain correctness properties (e. g., strong normalization in typed lambda calculus)

Types in Logic

• Many simple logics (e. g., propositional logics) are untyped. Even if there are functions (e. g., in first-order logic) the logic can be untyped (yet there exist typed versions of FOL).

• Many higher-order logics are typed, for much the same reasons than programming languages.

• Specification provided on the type level is unneccessary because it is expressible in the (untyped) logic itself. But type specification is convenient (especially when it is automatically checkable).

• In type theory there is a close link between types and logic.

(However, PVS is not based on type theory and does not exploit this.)

Types in PVS

In PVS, a type is a set of values. Questions:

• Which values contains a given type? • How can one construct these values? • How can one construct the types? • What purpose serve the types?

Constructing Types from Types

Basic builtin types:

• bool: two-element set of truth values TRUE, FALSE • nat: countable set of natural numbers 0, 1, 2, . . . • int: countable set of integers . . . , −2, −1, 0, 1, 2,. . . • rat: countable set of rational numbers, e. g., 1, 0.5,^1_3 , . . . • real: uncountable set of real numbers, e. g., 1, 0.33,^{1√}_3 , π, . . .

Constructing Types from Types

Tuples:

• type: [T_1, . . . , T_n] (n ≥ 1) • meaning: cartesian product T_1 × · · · × T_n • constructor: ( ) • destructors: ‘1, . . . , ‘n • examples:

[int] = int
(7, TRUE) : [int, bool]
(7, TRUE)‘1 = 7
(7, TRUE)‘2 = TRUE

Constructing Types from Types

Records: tuples with named components.

• type: [#a_1 : T_1, . . . , a_n : T_n#] (n ≥ 1) • meaning: cartesian product T_1 × · · · × T_n • constructor: (# #) • destructors: a_1, . . . , a_n • examples:

(# left := 7, right := TRUE #) : [# left: int, right: bool #]
(# left := 7, right := TRUE #) = (# right := TRUE, left := 7 #)
left((# left := 7, right := TRUE #)) = 7
right((# left := 7, right := TRUE #)) = TRUE

Constructing Types from Types

Functions:

• type: [T_1, . . . , T_n->T] (n ≥ 1) • meaning: function space T^{T1×···×Tn} • [T_1, . . . , T_n->T] = [[T_1, . . . , T_n]->T] • constructors: LAMBDA, RECURSIVE • destructor: function application • examples:

(LAMBDA x: x+1) : [int -> int]
(LAMBDA x: x+1)(7) = 7 + 1

Constructing Types from Types

Predicates: functions of type [T -> bool]

• examples:

pos(i: int): bool = i > 0

even(n: nat): RECURSIVE bool =
  IF n = 0 THEN TRUE
    ELSIF n = 1 THEN FALSE
      ELSE even(n-2)
  ENDIF
  MEASURE n

prime(n: nat): bool
AXIOM prime(n) = FORALL m: divides(m,n) IMPLIES m = 1 OR m = n

Constructing Types from Types

Predicates subtypes:

• type: {x: T | p(x)} or (p) where p: [T -> bool] • meaning: subset {x ∈ T | p(x)} of T • Predicate subtypes may be empty. • Type checking is undecidable. • Terms may have multiple types. • examples:

2: (pos)
2: (even)
2: (prime)

Constructing Types from Types

Dependent types: generalize tuple and function types.

• Types: set of all types • dependent pair: [x : A, B(x)] where B : A → Types • meaning: dependent product {⟨x, y⟩ ∈ A × ⋃ Types | y ∈ B(x)} • dependent function: [x : A → B(x)] where B : A → Types • meaning: dependent function space {f ∈ ( ⋃ Types)^A | ∀x ∈ A :

f(x) ∈ B(x)} • In PVS: For all x, B(x) must be a predicate subtype of some common supertype.

Constructing Types from Types

Dependent types example:

Informal specification: take takes a list xs and a natural number n and returns the prefix of length n of xs.

• No type dependencies:

– take: [list[int], nat -> list[int]] • Dependencies between arguments:

– take(xs: list[int], n: {i: nat | i <= length(xs)}): list[int] – take: [xs: list[int], {i: nat | i <= length(xs)} -> list[int]] • Dependencies between arguments and result:

– take: [xs: list[int], n: {i: nat | i <= length(xs)}

-> {zs: list[int] | length(zs) = n}]

Constructing Types from Nothing

• So far: Construct types from basic types via the type constructors (dependent) product, (dependent) function space and predicate subtype.

• Now: Synthesize an inductive type from no other type.

Inductive Types

Examples:

• One element type (constructor one, recognizer one?)

unit: DATATYPE
BEGIN
  one: one?
END unit

• Two element type (constructors tt, ff, recognizers tt?, ff?)

mybool: DATATYPE
BEGIN
  tt: tt?
  ff: ff?
END mybool

• General enumeration types (n constructors, n recognizers) 14

Inductive Types

Examples:

• Natural numbers (constructors zero, succ, destructor pred, recognizers zero?, succ?)

mynat: DATATYPE
BEGIN
  zero : zero?
  succ(pred: mynat): succ?
END mynat

• meaning: countable set of distinct constructor terms zero,

succ(zero), succ(succ(zero)), . . .

• PVS generates file mynat adt.pvs containing axioms defining – interaction between constructors, destructors and recognizers, – the subterm relation and generic measure functions 15

Inductive Types

Examples:

• Lists of booleans (constructors null, cons, destructors car, cdr, recognizers emptylist?, nonemptylist?)

boollist: DATATYPE
BEGIN
  null: emptylist?
  cons(car: bool, cdr: boollist): nonemptylist?
END boollist

• meaning: countable set of distinct constructor terms null,

cons(FALSE,null), cons(TRUE,null), cons(FALSE,cons(FALSE,null)),
cons(FALSE,cons(TRUE,null)), cons(TRUE,cons(FALSE,null)),
cons(TRUE,cons(TRUE,null)), . . .

Inductive Types

Examples:

• Terms for integers (constructors zero, succ, pred)

myint_term: DATATYPE
BEGIN
  zero : zero?
  succ(predofsucc: myint_term): succ?
  pred(succofpred: myint_term): pred?
END myint_term

• meaning of myint term: set of constructor terms zero,

succ(zero), pred(zero), succ(succ(zero)), succ(pred(zero)),
pred(succ(zero)), pred(pred(zero)), . . .

Inductive Types

Example continued:

• Identify semantically equivalent terms through axioms

myint: THEORY
BEGIN
  IMPORTING myint_term
  myint: TYPE = myint_term
  pred_succ: AXIOM FORALL (x: myint): pred(succ(x)) = x
  succ_pred: AXIOM FORALL (x: myint): succ(pred(x)) = x
END myint

• meaning of myint: set of equivalence classes of constructor terms

Structural Recursion over Inductive Types

Examples:

• Recursion over an enumeration type

not(b: mybool): mybool =
  CASES b OF
    tt: ff,
    ff: tt
  ENDCASES

• Recursion over an infinite inductive type

add(m, n: mynat): RECURSIVE mynat =
  CASES m OF
    zero: n,
    succ(k): succ(add(k,n))
  ENDCASES
MEASURE m BY <<

19

Declaring Uninterpreted Types

Uninterpreted types are essentially free type variables. They can stand for any type which is expressible in PVS.

• declaration of uninterpreted type T_1:

T_1: TYPE

• nonempty uninterpreted types T_2 and T_3:

T_2: NONEMPTY_TYPE
T_3: TYPE+

• uninterpreted subtypes T_4 and T_5 of T_1:

T_4: TYPE FROM T_1
T_4: NONEMPTY_TYPE FROM T_1

References

1. PVS Language Reference. Chapter 4 and 8.

2. Sam Owre, Natarajan Shankar. The Formal Semantics of PVS.

Technical Report CSL-97-2R. 1997.